# Parameter Path Estimation in Unstable Environments: The `tvpreg` Command

Atsushi Inoue[1], Barbara Rossi[2], Yiru Wang[3], Lingyun Zhou[4]

[1]Vanderbilt University

[2]ICREA-Universitat Pompeu Fabra, EUI, Barcelona School of Economics, and CREI

[3]University of Pittsburgh

[4]Tsinghua University

2024 Stata Economics Symposium

# Motivation

- Time series models often exhibit parameter instability.
- Documenting the time variation is useful:
  - The path estimators help describe the potential sources of instability;
  - The model has insights on how macroeconomics variables respond to structural shocks at different periods;
  - The end point of the parameter path is useful for forecasting.
- Small parameter variation is difficult to detect but empirically common:
  - Typical tests would detect the instability with probability $< 1$ even in the limit, see Müller and Petalas' (2010)

# Motivation Example 1: IRF

- Impulse response functions (IRFs) are key objects of interest in empirical macroeconomic analysis.
  - Responses of target variable $(Y_{i,t})$ due to exogenous policy change, at horizons $h = 1, 2, \ldots$.
  - Response of the economy to monetary policy, to fiscal policy, government spending multiplier, etc

# What is Impulse Response Function (IRF)?

- Let $Y_t$, a $(K \times 1)$ vector of macroeconomic variables, be written in terms of current and past shocks $\epsilon_t$ in a structural moving average representation:

$$Y_t = \Theta(L)\epsilon_t = \Theta_0\epsilon_t + \Theta_1\epsilon_{t-1} + ... + \Theta_h\epsilon_{t-h} + ...$$

  - The coefficients of $\Theta(L)$ are the structural IRFs.
  - The causal effect of a unit increase in $\epsilon_1$ on the value of the second variable, $Y_2$, $h$ periods hence, is

$$\begin{aligned}\Theta_{h,21} =& E[Y_{2,t+h}|\epsilon_{1,t} = 1, \epsilon_{2:K,t}, \epsilon_s, s < t] \\ &- E[Y_{2,t+h}|\epsilon_{1,t} = 0, \epsilon_{2:K,t}, \epsilon_s, s < t], \quad h = 0, 1, \ldots\end{aligned}$$

# How do researchers get IRFs?

- VAR-based IRFs: recursively iterating VARs

$$Y_t = B_1 Y_{t-1} + \ldots + B_p Y_{t-p} + U_t$$

- Local projections (LPs): estimate IRF directly

$$Y_{2,t+h} = \Theta_{h,21}\epsilon_{1,t} + \gamma_h' W_t + u_{2,t+h}^h, \qquad h = 0, 1, \cdots$$

How to estimate IRFs in VAR and LP in the presence of instabilities?

# TVP-LP and TVP-VAR

In other words, we are interested in:

- the time-varying local projection (TVP-LP) regression

$$Y_{2,t+h} = \Theta_{h,21,t}\epsilon_t + \gamma'_{h,t}W_t + u^h_{t+h}, \qquad h = 0, 1, \cdots ,$$

  where $W_t$ are control variables and the residual $u^h_{t+h}$ is serially correlated with variance $\sigma^2_{u,t}$.

- the time-varying parameter VAR (TVP-VAR) regression

$$Y_t = B_{1,t}Y_{t-1} + \ldots + B_{p,t}Y_{t-p} + U_t,$$

  where the error term $U_t$ has covariance matrix $\Sigma_{U,t}$.

# Motivation Example 2: Phillips curve

The Phillips curve is a structural relationship describing how inflation changes as a function of real activity:

$$\pi_t = c + \gamma_f E_t(\pi_{t+1}) + \gamma_b \pi_{t-1} + \lambda x_t + u_t,$$

where $\pi_t$ = inflation, $x_t$ = real marginal cost measure (unemployment), $E_t(.)$ = conditional expectations at time $t$.

There is substantial evidence of time-variations in the parameters.

The time-varying parameter counterpart:

$$\pi_t = c_t + \gamma_{f,t} E_t(\pi_{t+1}) + \gamma_{b,t} \pi_{t-1} + \lambda_t x_t + u_t,$$

# WAR minimizing path estimators

Methodologically, we consider Müller and Petalas' (2010) path estimator for the parameters of interest, which

- allows for smooth and flexible time-variation;
- allows for time-variation in both coefficients and variances;
- is a weighted average risk (WAR) minimizing path estimator;
- simplifies the estimation procedures in the presence of instabilities.

Related papers:

- Müller-Petalas (2010): path estimator
- Inoue, Rossi, and Wang (2024b): TVP-LP and TVP-LP-IV
- Inoue, Rossi, and Wang (2024a): TVP-weak IV

# The `tvpreg` Command

The `tvpreg` command offers to capture time-varying parameters in regression models in Stata environment.

- Implements the Weighted Average Risk (WAR) minimizing path estimators.
- Visualizes and stores time-varying coefficients' path estimates with postestimation commands.

# Econometric Framework

# Stable Environment

Consider the following general model in a stable environment:

$$\mathbf{y}_t = \mathbf{B}\mathbf{X}_t + \mathbf{e}_t,$$

where

- $\mathbf{y}_t$: Dependent variable
- $\mathbf{X}_t$: Explanatory variables
- $\mathbf{B}$: Parameter matrix
- $\mathbf{e}_t$: Error term with mean zero and covariance matrix $\mathbf{\Sigma}_e$

Let $\boldsymbol{\theta}$ include all the parameters.

- $\boldsymbol{\theta} = (\mathbf{B}, \ln \sigma_e)$ if it is a univariate regression;
- $\boldsymbol{\theta} = (vec(\mathbf{B}')', vech(\mathbf{\Sigma}_e)')'$ if it is a multivariate regression.

Let $f(\mathbf{y}_t|\mathbf{X}_t, \boldsymbol{\theta})$ denote the conditional pdf for $\mathbf{y}_t$.
Log-Likelihood Function:

$$\sum_{t=1}^{T} \ell_t(\boldsymbol{\theta}), \quad \ell_t(\boldsymbol{\theta}) = \ln f(\mathbf{y}_t|\mathbf{X}_t, \boldsymbol{\theta})$$

# Stable Environment

- The gradient $\mathbf{s}_t(\boldsymbol{\theta})$ and the negative Hessian matrix $\mathbf{H}_t(\boldsymbol{\theta})$ are summarizing the information of the constant parameter estimator $\widehat{\boldsymbol{\theta}}$.
- In many scenarios (e.g., local projection discussed in Inoue, Rossi, and Wang 2024b), the estimated parameters are asymptotically normally distributed:

$$\sqrt{T}(\widehat{\boldsymbol{\theta}} - \boldsymbol{\theta}_0) \Rightarrow \mathcal{N}(\mathbf{0}, \mathbf{S})$$

where $\widehat{\mathbf{S}} = \widehat{\mathbf{H}}^{-1}\widehat{\mathbf{V}}\widehat{\mathbf{H}}^{-1}$ is the sandwich asymptotic variance with $\widehat{\mathbf{H}} = \frac{1}{T}\sum_{t=1}^{T}\mathbf{H}_t(\widehat{\boldsymbol{\theta}})$ and $\widehat{\mathbf{V}} = \frac{1}{T}\sum_{t=1}^{T}\mathbf{s}_t(\widehat{\boldsymbol{\theta}})\mathbf{s}_t(\widehat{\boldsymbol{\theta}})'$ if the score is i.i.d. (otherwise, HAC estimators can be used)

# Unstable Environments

Consider the following counterpart in an unstable environment:

$$\mathbf{y}_t = \mathbf{B}_t \mathbf{X}_t + \mathbf{e}_t,$$

where the error term $\mathbf{e}_t$ has a zero mean, and a time-varying covariance matrix $\mathbf{\Sigma}_{e,t}$.

**Parameter Path:** $\{\boldsymbol{\theta}_t\}_{t=1}^T = \{\boldsymbol{\theta} + \boldsymbol{\delta}_t\}_{t=1}^T$ with $\sum_{t=1}^T \boldsymbol{\delta}_t = 0$

- allows for local time variation in both the slope and variance coefficients
- its variability is assumed of the order of magnitude $T^{-1/2}$ (satisfying Condition 2 in Müller and Petalas, 2010)
- can accommodate a wide range of parameter instabilities, including a random-walk type variability occurring in either the full sample or in specific subsamples, piece-wise constant paths with finitely many breaks, etc.
- TVP-VAR, TVP-LP, ...

# Unstable Environments

The system has the same log-likelihood function $\sum_{t=1}^{T} \ell_t(.)$ but with time-varying parameters $\{\boldsymbol{\theta}_t\}_{t=1}^{T}$.

A quadratic approximation to log-likelihood. The sample information about the parameter path $\{\boldsymbol{\theta}_t\}_{t=1}^{T} = \{\boldsymbol{\theta} + \boldsymbol{\delta}_t\}_{t=1}^{T}$ can be summarized using the pseudo model, see Müller and Petalas (2010):

$$\widehat{\boldsymbol{\theta}} = \boldsymbol{\theta} + T^{-1/2}\widehat{\mathbf{S}}\mathbf{v}_0,$$

$$\widehat{\mathbf{H}}\widehat{\mathbf{V}}^{-1}\mathbf{s}_t(\widehat{\boldsymbol{\theta}}) = \widehat{\mathbf{S}}^{-1}\boldsymbol{\delta}_t + \mathbf{v}_t, \quad t = 1, ..., T,$$

with $\mathbf{v}_t \sim \mathcal{N}(\mathbf{0}, \widehat{\mathbf{H}})$.

# The Estimator

The parameter path $\widehat{\boldsymbol{\theta}}_t, t = 1, 2, .., T$, minimizes the weighted average risk (Müller and Petalas, 2010, Inoue et al., 2024b), where the weighting is over alternative parameter paths.

The optimal parameter path estimator will assume a weighting function such that $\boldsymbol{\theta}_t$ is a multivariate Gaussian random walk with a small variance (Canova, 1983, Cogley-Sargent, 2005, Primiceri, 2005):

$$\boldsymbol{\theta}_t = \boldsymbol{\theta}_{t-1} + \epsilon_t,$$

where $\epsilon_t$ is an i.i.d. disturbance with variance $(1/T)\sigma^2_{\epsilon,t}$.

Theoretical justification of the methodology: optimality properties (minimizes WAR) is shown in Müller and Petalas (2010). The propositions in Inoue, Rossi, and Wang (2024b) show optimality in TVP-LP(-IV) setting.

# Step-by-Step Implementation

**Step 1.** For $t = 1, \ldots, T$, let $\widetilde{\mathbf{x}}_t$ and $\widetilde{\mathbf{y}}_t$ be all the elements of $\widehat{\mathbf{H}}^{-1}\mathbf{s}_t(\widehat{\boldsymbol{\theta}})$ and $\widehat{\mathbf{H}}\widehat{\mathbf{V}}^{-1}\mathbf{s}_t(\widehat{\boldsymbol{\theta}})$, respectively.

**Step 2.** For $c_i \in \mathbf{C} = \{c_0, c_1, c_2, \ldots, c_{n_G}\}$, $i = 0, 1, \ldots, n_G$ and $c_0 = 0$, compute

(a) $r_i = 1 - \frac{c_i}{T}$, $\boldsymbol{\zeta}_{i,1} = \widetilde{\mathbf{x}}_1$, and $\boldsymbol{\zeta}_{i,t} = r_i \boldsymbol{\zeta}_{i,t-1} + \widetilde{\mathbf{x}}_t - \widetilde{\mathbf{x}}_{t-1}$, $t = 2, \ldots, T$;
(b) the residuals $\{\widetilde{\boldsymbol{\zeta}}_{i,t}\}_{t=1}^{T}$ of a linear regression of $\{\boldsymbol{\zeta}_{i,t}\}_{t=1}^{T}$ on $\{r_i^{t-1}\mathbf{I}_q\}_{t=1}^{T}$;
(c) $\overline{\boldsymbol{\zeta}}_{i,T} = \widetilde{\boldsymbol{\zeta}}_{i,T}$, and $\overline{\boldsymbol{\zeta}}_{i,t} = r_i \overline{\boldsymbol{\zeta}}_{i,t+1} + \widetilde{\boldsymbol{\zeta}}_{i,t} - \widetilde{\boldsymbol{\zeta}}_{i,t+1}$, $t = 1, \ldots, T-1$;
(d) $\{\widehat{\boldsymbol{\theta}}_{i,t}\}_{t=1}^{T} = \{\widehat{\boldsymbol{\theta}} + \widetilde{\mathbf{x}}_t - r_i \overline{\boldsymbol{\zeta}}_{i,t}\}_{t=1}^{T}$;
(e) $qLL(c_i) = \sum_{t=1}^{T}(r_i \overline{\boldsymbol{\zeta}}_{i,t} - \mathbf{x}_t)'\widetilde{\mathbf{y}}_t$ and $\widetilde{w}_i = \sqrt{T(1-r_i^2)r_i^{T-1}/((1-r_i^{2T}))} exp[-\frac{1}{2}qLL(c_i)]$ (set $\widetilde{w}_0 = 1$).

**Step 3.** Compute $w_i = \widetilde{w}_i / \sum_{j=0}^{n_G} \widetilde{w}_j$.

**Step 4.** The parameter path estimator is given by $\{\widehat{\boldsymbol{\theta}}_t\}_{t=1}^{T} = \{\sum_{i=0}^{n_G} w_i \widehat{\boldsymbol{\theta}}_{i,t}\}_{t=1}^{T}$.

**Step 5.** With the weighting functions for $\{\boldsymbol{\delta}_t\}_{t=1}^{T}$ and $\boldsymbol{\theta}$ interpreted as priors from a Bayesian perspective, the approximate posterior for $\boldsymbol{\theta}_t$ is a mixture of multivariate normals $\mathcal{N}\left(\widehat{\boldsymbol{\theta}}_{i,t}, T^{-1}\widehat{\mathbf{S}}\kappa_t(c_i)\right)$, $i = 0, \ldots, n_G$ with mixing probabilities $w_i$ where $\widehat{\mathbf{S}} = \widehat{\mathbf{H}}^{-1}\widehat{\mathbf{V}}\widehat{\mathbf{H}}^{-1}$, $\kappa_t(c) = \frac{c(1+e^{2c}+e^{2ct/T}+e^{2c(1-t/T)})}{2e^{2c}-2}$, and $\kappa_t(0) = 1$. Following the results, the mixture of normals is approximated by $N(\widehat{\boldsymbol{\theta}}_t, \boldsymbol{\Omega}_{\theta,t})$ where $\boldsymbol{\Omega}_{\theta,t} = \sum_{i=0}^{n_G} w_i \left(T^{-1}\widehat{\mathbf{S}}\kappa_t(c_i) + (\widehat{\boldsymbol{\theta}}_{i,t} - \widehat{\boldsymbol{\theta}}_t)(\widehat{\boldsymbol{\theta}}_{i,t} - \widehat{\boldsymbol{\theta}}_t)'\right)$. Thus, the WAR minimizing interval estimator $\left[\widehat{\boldsymbol{\theta}}_{j,t} - 1.96\sqrt{\boldsymbol{\Omega}_{\theta,jj,t}}, \widehat{\boldsymbol{\theta}}_{j,t} + 1.96\sqrt{\boldsymbol{\Omega}_{\theta,jj,t}}\right]$ is the 95% equal-tailed approximate posterior probability interval for $\boldsymbol{\theta}_{j,t}$, the $j$-th element of $\boldsymbol{\theta}$ at time $t$; where $\widehat{\boldsymbol{\theta}}_{j,t}$ is the $j$-th element of $\widehat{\boldsymbol{\theta}}_t$ and $\boldsymbol{\Omega}_{\theta,jj,t}$ is the $(j,j)$ element of $\boldsymbol{\Omega}_{\theta,t}$.

# Endogeneity

Suppose in our model:

$$\mathbf{y}_t = \mathbf{B}_t \mathbf{X}_t + \mathbf{e}_t = \mathbf{B}_{x,t} \mathbf{x}_t + \mathbf{B}_{z1,t} \mathbf{z}_{1,t} + \mathbf{e}_t$$

**Challenge:** $\mathbf{x}_t$ is endogenous.

**Solution:** Introduce external instruments $\mathbf{z}_{2,t}$ satisfying:

- Relevance: $E[\mathbf{x}_t \mathbf{z}'_{2,t}]$ has full row rank
- Exogeneity: $E[\mathbf{z}_{2,t} \mathbf{e}'_t] = \mathbf{0}$

# Time-Varying Parameter IV Model (TVP-IV)

Then we can write the following multivariate system:

$$\begin{bmatrix} \mathbf{x}_t \\ \mathbf{y}_t \end{bmatrix} = \begin{bmatrix} \mathbf{\Pi}_{2,t} & \mathbf{\Pi}_{1,t} \\ \mathbf{B}_{x,t}\mathbf{\Pi}_{2,t} & \mathbf{B}_{x,t}\mathbf{\Pi}_{1,t} + \mathbf{B}_{z1,t} \end{bmatrix} \begin{bmatrix} \mathbf{z}_{2,t} \\ \mathbf{z}_{1,t} \end{bmatrix} + \begin{bmatrix} \boldsymbol{\nu}_{1,t} \\ \boldsymbol{\nu}_{2,t} \end{bmatrix}$$
$$= \begin{bmatrix} \mathbf{\Pi}_t \\ \mathbf{B}_{x,t}\mathbf{\Pi}_t + \begin{bmatrix} 0 & \mathbf{B}_{z1,t} \end{bmatrix} \end{bmatrix} \begin{bmatrix} \mathbf{z}_{2,t} \\ \mathbf{z}_{1,t} \end{bmatrix} + \begin{bmatrix} \boldsymbol{\nu}_{1,t} \\ \boldsymbol{\nu}_{2,t} \end{bmatrix},$$

where $\boldsymbol{\nu}_t = (\boldsymbol{\nu}'_{1,t}, \boldsymbol{\nu}'_{2,t})'$ has zero mean and covariance matrix $\mathbf{\Sigma}_{\nu,t}$.

Let $\boldsymbol{\theta}_t = (\text{vech}(\mathbf{\Pi}')', \text{vech}(\mathbf{B}'_x)', \text{vech}(\mathbf{B}'_{z1})', \text{vech}(\mathbf{\Sigma}_{\nu,t})')'$ include all the time-varying parameters of interest. Let $f(\widetilde{\mathbf{y}}_t | \mathbf{z}_{2,t}, \mathbf{z}_{1,t}, \boldsymbol{\theta}_t)$ denote a family of conditional density functions for $\widetilde{\mathbf{y}}_t = (\mathbf{x}'_t, \mathbf{y}'_t)'$.

Once the scores and hessians are constructed, we can get the WAR minimizing path estimator $\widehat{\boldsymbol{\theta}}_t, t = 1, 2, .., T$.

# Handling Weak Instruments

**Challenge:** Weak instruments prevent direct estimation of the structural parameters in the TVP-IV model.

**Idea:** Although we cannot estimate the structural parameters directly, we can still estimate the reduced-form parameters.

Rewrite the model in terms of reduced-form parameters:

$$\begin{bmatrix} \mathbf{x}_t \\ \mathbf{y}_t \end{bmatrix} = \begin{bmatrix} \mathbf{\Pi}_t \\ \mathbf{\Gamma}_t \end{bmatrix} \begin{bmatrix} \mathbf{z}_{2,t} \\ \mathbf{z}_{1,t} \end{bmatrix} + \begin{bmatrix} \boldsymbol{\nu}_{1,t} \\ \boldsymbol{\nu}_{2,t} \end{bmatrix}$$

where the reduced-form parameters $\mathbf{\Gamma}_t = \mathbf{B}_{x,t}\mathbf{\Pi}_t + \begin{bmatrix} \mathbf{0} & \mathbf{B}_{z1,t} \end{bmatrix}$ is function of structural parameters.

We first estimate the reduced-form parameters $\mathbf{\Pi}_t$ and $\mathbf{\Gamma}_t$, then recover structural parameters $\mathbf{B}_{x,t}$ and $\mathbf{B}_{z1,t}$.

# Recovering Structural Parameters

**Procedure:**

- Let $\boldsymbol{\theta}_t^{rf} = (\text{vech}(\boldsymbol{\Pi}')', \text{vech}(\boldsymbol{\Gamma}')', \text{vech}(\boldsymbol{\Sigma}_{\nu,t})')'$ include all the time-varying parameters of interest.
- Obtain the WAR minimizing path/interval estimate for the reduced-form parameter $\boldsymbol{\theta}_t^{rf}$, thus the WAR minimizing path/interval estimate for $\widehat{\boldsymbol{\Pi}}_t$ and $\widehat{\boldsymbol{\Gamma}}_t$.
- Recover structural parameters $\widehat{\mathbf{B}}_{x,t}$, $\widehat{\mathbf{B}}_{z1,t}$.

# Parameter Definitions

**Parameters of Interest** $(\theta_t)$:

- Case (i): No covariance decomposition

$$\theta_t = (\boldsymbol{\Pi}_{1,:,t} \cdots \boldsymbol{\Pi}_{n_x,:,t}, \mathbf{B}_{x,1,:,t} \cdots \mathbf{B}_{x,n_y,:,t}, \mathbf{B}_{z1,1,:,t} \cdots \mathbf{B}_{z1,n_y,:,t}, \text{vech}(\boldsymbol{\Sigma}_{\nu,t}))'$$

- Case (ii): With triangular reduction of $\boldsymbol{\Sigma}_{\nu,t}$ (e.g., TVP-VAR)

$$\theta_t = (\boldsymbol{\Pi}_{1,:,t} \cdots \boldsymbol{\Pi}_{n_x,:,t}, \mathbf{B}_{x,1,:,t}, \mathbf{B}_{z1,1,:,t}, \mathbf{a}_t', \ln \boldsymbol{\sigma}_t')'$$

where $\mathbf{a}_t$ contains the elements below the diagonal of $\mathbf{A}_t$ and $\boldsymbol{\sigma}_t = diag(\boldsymbol{\Sigma}_{\epsilon,t})$. Here, $\mathbf{A}_t$ is a lower triangular matrix with diagonal elements equal to one and $\boldsymbol{\Sigma}_{\epsilon,t}$ is a diagonal matrix such that $\mathbf{A}_t\boldsymbol{\Sigma}_{\nu,t}\mathbf{A}_t' = \boldsymbol{\Sigma}_{\epsilon,t}^2$.

# The tvpreg command

# Command Syntax

The syntax of the **tvpreg** command is as follows:

**tvpreg** *varlist_dep varlist1* (*varlist2* = *varlist_iv*) [*if*] [*in*] [, *estimator*
**cmatrix**(*matname*) **nhorizon**(*numlist*) **cum** **slope** **cholesky** **nwlag**(#)
**noconstant** **ny**(#) **varlag**(*numlist*) **ndraw**(#) **getband** **level**(*clevel*)
**nodisplay** **plotcoef**(*namelist*) **plotvarirf**(*namelist*) **plotnhorizon**(*numlist*)
**plotconst** **period**(*varname*) **movavg**(#) **noci** **title**(*tinfo*)
**ytitle**(*axis_title*) **xtitle**(*axis_title*) **tvplegend**(*string*) **constlegend**(*string*)
**bandlegend**(*string*) **shadelegend**(*string*) **periodlegend**(*namelist*) **nolegend**
**scheme**(*schemename*) **tvpcolor**(*colorstyle*) **constcolor**(*colorstyle*)
**name**(*name_option*) ]

# Some Important Options

*estimator* specifies the estimation method for the constant parameter estimate which is used to evaluate the scores and Hessians.

- `ols`: Ordinary least squares (default if no instruments)
- `newey`: HAC estimation
- `2sls`: Two-stage least squares (default with instruments)
- `gmm`: Generalized method of moments
- `weakiv`: Weak-instrument estimation using OLS on reduced-form regression
- `var`: Vector autoregressive model

# Some Important Options

cmatrix(*matname*) specifies the **C** grid for $c_i$. This option controls the magnitude of time variation allowed in the parameters: larger $c_i$ values allow for larger magnitude of time variation. The default is a scalar $c_i$ with the **C** grid $0:5:50$. We allow $c_i$ to be either a scalar or a vector: a vector $c_i$ allows different magnitudes of time-variation in different blocks of the parameters. When $c_i$ is a scalar, **C** is a row vector; when $c_i$ is a vector, **C** is a matrix whose number of rows should equal the number of parameters.

# Some Important Options

- `nhorizon(numlist)`: List of horizons in local projections or VAR; default is `nhorizon(0)`
- `lplagged`: Holds endogenous variables at current period across horizons
- `cumulative`: Cumulates variables over horizons
- `slope`: Only slope parameters vary over time
- `cholesky`: Logarithm of standard deviation (univariate) or triangular reduction of covariance (multivariate)

# Additional Options

- `nwlag(num)`: Sets number of lags for calculating the long run variance of scores, default depends on estimator
- `nocons`: Suppresses constant term
- `ny(num)`: Number of dependent variables (default is 1)
- `varlag(numlist)`: Sets lags for VAR models
- `ndraw(num)`: Sets draws for confidence band estimation
- `getband`: Generates confidence bands
- `level(clevel)`: Sets confidence level (default 95)
- `nodisplay`: Suppresses text output

# Plotting Options

- `plotcoef(namelist)`: Specifies coefficients to plot
- `plotvarirf(namelist)`: Specifies impulse responses for VAR
- `plotnhorizon(numlist)`: Horizons to plot (subset of `nhorizon`)
- `plotconst`: Includes constant parameter estimate
- `period(varname)`: Highlights specified time points
- `movavg(num)`: Sets degree for moving average smoothing
- `noci`: Suppresses confidence band in plots

# Additional Plotting Customization

- `title(tinfo)`: Sets graph title
- `ytitle(axis_title)`: Y-axis title (default: "Parameter")
- `xtitle(axis_title)`: X-axis title (default: "Time" or "Horizons")
- `tvplegend(string)`: Legend for time-varying parameter estimate
- `constlegend(string)`: Legend for constant parameter estimate
- `bandlegend(string)`: Legend for confidence band
- `shadedlegend(string)`: Legend for shaded regions (for `period`)
- `periodlegend(namelist)`: Legend for estimates in specified time periods

# Graph Appearance Options

- `nolegend`: Suppresses legend
- `scheme(schemename)`: Sets graph scheme
- `tvpcolor(colorstyle)`: Sets color of time-varying estimate (default: green)
- `constcolor(colorstyle)`: Sets color of constant estimate (default: black)
- `name(name_option)`: Sets graph name (default: "tvpreg")

# List of specifications considered by `tvpreg`

| Model | Specification | Code |
| --- | --- | --- |
| TVP-REG | $\mathbf{y}_t = \mathbf{B}_t \mathbf{X}_t + \mathbf{e}_t$ | `tvpreg y x, ...` |
| TVP-IV | $\mathbf{y}_t = \mathbf{B}_{x,t}\mathbf{x}_t + \mathbf{B}_{z1,t}\mathbf{z}_{1,t} + \mathbf{e}_t$, where $\mathbf{z}_{2,t}$ is the external instrument for the endogenous variable $\mathbf{x}_t$ | `tvpreg y z1 (x = z2), ...` |
| TVP-weak IV | $\mathbf{y}_t = \mathbf{B}_{x,t}\mathbf{x}_t + \mathbf{B}_{z1,t}\mathbf{z}_{1,t} + \mathbf{e}_t$, where $\mathbf{z}_{2,t}$ is the external instrument for the endogenous variable $\mathbf{x}_t$ and the instrument is weak | `tvpreg y z1 (x = z2), weakiv ...` |

# List of specifications considered by tvpreg (continued)

| Model | Specification | Code |
|---|---|---|
| TVP-VAR | $\mathbf{B}_t(L)\mathbf{y}_t = \mathbf{c}_t + \mathbf{e}_t = \mathbf{c}_t + \Theta_{0,t}\boldsymbol{\epsilon}_t$ with recursive identification | `tvpreg ylist, var varlag(1/L) chol nhor(0/H)` |
| TVP-LP-1 | $\mathbf{y}_{t+h} = \mathbf{B}_{h,t+h}\mathbf{X}_t + \mathbf{e}_{t+h}$, where $\mathbf{X}_t$ includes lagged variables | `tvpreg y x, nhor(0/H) ...` |
| TVP-LP-2 | $\sum_{j=0}^{h} \mathbf{y}_{t+j} = \mathbf{B}_{h,t+h}\mathbf{X}_t + \mathbf{e}_{t+h}$, where $\mathbf{X}_t$ includes lagged variables | `tvpreg y x, nhor(0/H) cum ...` |
| TVP-LP-IV-1 | $\mathbf{y}_{t+h} = \mathbf{B}_{x,t}\mathbf{x}_{t+h} + \mathbf{B}_{z1,t}\mathbf{z}_{1,t} + \mathbf{e}_t$, where $\mathbf{z}_{2,t}$ is the external instrument for the endogenous variable $\mathbf{x}_{t+h}$ | `tvpreg y z1 (x = z2), nhor(0/H) ...` |
| TVP-LP-IV-2 | $\mathbf{y}_{t+h} = \mathbf{B}_{x,t}\mathbf{x}_t + \mathbf{B}_{z1,t}\mathbf{z}_{1,t} + \mathbf{e}_t$, where $\mathbf{z}_{2,t}$ is the external instrument for the endogenous variable $\mathbf{x}_t$ | `tvpreg y z1 (x = z2), nhor(0/H) lplagged ...` |
| TVP-LP-IV-3 | $\sum_{j=0}^{h} \mathbf{y}_{t+j} = \mathbf{B}_{x,t} \sum_{j=0}^{h} \mathbf{x}_{t+j} + \mathbf{B}_{z1,t}\mathbf{z}_{1,t} + \mathbf{e}_t$, where $\mathbf{z}_{2,t}$ is the external instrument for the endogenous variable $\sum_{j=0}^{h} \mathbf{x}_{t+j}$ | `tvpreg y z1 (x = z2), nhor(0/H) cum ...` |
| TVP-LP-IV-4 | $\sum_{j=0}^{h} \mathbf{y}_{t+j} = \mathbf{B}_{x,t}\mathbf{x}_t + \mathbf{B}_{z1,t}\mathbf{z}_{1,t} + \mathbf{e}_t$, where $\mathbf{z}_{2,t}$ is the external instrument for the endogenous variable $\mathbf{x}_t$ | `tvpreg y z1 (x = z2), nhor(0/H) lplagged cum` |

# Implementation Examples

# Summary of the empirical applications

| Model | Applications | Reference |
|---|---|---|
| TVP-VAR | Monetary policy shock | Primiceri (2005) |
| TVP-LP | Fiscal shock | Ramey and Zubairy (2018), Inoue, Rossi, and Wang (2024b) |
| TVP-LP-IV | Fiscal multiplier | Ramey and Zubairy (2018), Inoue, Rossi, and Wang (2024b) |
| TVP-weak IV | Phillips curve | Galí and Gertler (1999), Inoue, Rossi, and Wang (2024a) |

For each application, the implementation follows five steps:

(1) Importing the data;

(2) Creating the **C** grid;

(3) Estimating the model using `tvpreg`;

(4) Plotting the estimation results using `tvpplot`;

(5) Saving the estimation results as new variables using `predict`.

# Example 1: TVP-VAR

# Empirical Setup

Primiceri (2005): structural VAR for the U.S. economy

- Inflation ($\pi_t$), Unemployment ($u_t$), Short-term interest rate ($i_t$)
- Sample Period: 1953Q1 to 2006Q3

**Goal:** To estimate time-varying IRFs through a VAR-based model.

The TVP-VAR model takes the form:

$$\begin{bmatrix} \pi_t \\ u_t \\ i_t \end{bmatrix} = \mathbf{B}_{1,t} \begin{bmatrix} \pi_{t-1} \\ u_{t-1} \\ i_{t-1} \end{bmatrix} + \ldots + \mathbf{B}_{p,t} \begin{bmatrix} \pi_{t-p} \\ u_{t-p} \\ i_{t-p} \end{bmatrix} + \mathbf{c}_t + \mathbf{e}_t$$

$$\mathbf{B}_t(L)\mathbf{y}_t = \mathbf{c}_t + \mathbf{e}_t = \mathbf{c}_t + \Theta_{0,t}\epsilon_t,$$

where:

- $\mathbf{y}_t = (\pi_t, u_t, i_t)'$, $\mathbf{B}_t(L) = \mathbf{I} - \mathbf{B}_{1,t}L - \cdots - \mathbf{B}_{p,t}L^p$
- $\mathbf{e}_t = \boldsymbol{\Theta}_{0,t}\epsilon_t$, with $\boldsymbol{\Sigma}_{e,t} = \text{cov}(\mathbf{e}_t)$
- The TVP-VAR impulse responses can be obtained by $\mathbf{B}_t(L)^{-1}\boldsymbol{\Theta}_{0,t}$.

# Cholesky Decomposition

To identify $\mathbf{\Theta}_{0,t}$, we impose a unit effect normalization such that the diagonal elements of $\mathbf{\Theta}_{0,t}$ are equal to 1.

Also, let

$$\mathbf{A}_t \mathbf{\Sigma}_{e,t} \mathbf{A}_t' = \mathbf{\Sigma}_{\epsilon,t}^2$$

where:

$$\mathbf{A}_t = \begin{bmatrix} 1 & 0 & 0 \\ a_{21,t} & 1 & 0 \\ a_{31,t} & a_{32,t} & 1 \end{bmatrix}, \quad \mathbf{\Sigma}_{\epsilon,t} = \begin{bmatrix} \sigma_{1,t} & 0 & 0 \\ 0 & \sigma_{2,t} & 0 \\ 0 & 0 & \sigma_{3,t} \end{bmatrix}$$

# Estimation

- Let $\boldsymbol{\theta}_t = (vec(\mathbf{B}_t')', \mathbf{a}_t', \ln \boldsymbol{\sigma}_t')'$ denote the vector containing all the time-varying parameters in the TVP-VAR, where $\mathbf{B}_t = [\mathbf{B}_{1,t}, ..., \mathbf{B}_{p,t}, \mathbf{c}_t]$, $\mathbf{a}_t$ contains the elements below the diagonal of $\mathbf{A}_t$, and $\boldsymbol{\sigma}_t = diag(\boldsymbol{\Sigma}_{\epsilon,t})$.
- Obtain WAR minimizing path estimators $\widehat{\boldsymbol{\theta}}_t$
- Calculate time-varying impulse responses $\widehat{IRF}_t = \widehat{\mathbf{B}}_t(L)^{-1}\widehat{\boldsymbol{\Theta}}_{0,t}$
- Generate confidence bands via sampling

# Implementation in Stata

**Step 1:** Import the data

```
.
. * Estimator I: TVP-VAR (Monetary Policy)
. // Step 1:  Import the data
. use data_MP.dta, clear

. tsset time

Time variable: time, 1953q1 to 2006q3
        Delta: 1 quarter
```

# Implementation in Stata

**Step 2:** Create the C grid
Example: $\mathbf{C}_B = \{0, 0.6, \ldots, 3\}$, $\mathbf{C}_a = \{0, 1.2, \ldots, 6\}$, $\mathbf{C}_{\ln\sigma} = \{0, 3, \ldots, 15\}$

```
. // Step 2: Create the C grid
. * Input correct # of parameters
. mata: ny = 3; nlag = 2; ncons = 1

. * Input C grids for different blocks of parameters
. mata: cB = (0.6*(0::5))'; ca = (1.2*(0::5))'; cl = (3*(0::5))'

. * Automatic calculations
. mata: nB = ny * (ny * nlag + ncons)

. mata: na = ny * (ny - 1) / 2; nl = ny

. mata: ncB = cols(cB); nca = cols(ca); ncl = cols(cl)

. mata: cB = cB # J(1,nca*ncl,1)

. mata: ca = J(1,ncB,1) # ca # J(1,ncl,1)

. mata: cl = cl = J(1,ncB*nca,1) # cl

. mata: cmat = (J(nB,1,1) # cB) \ (J(na,1,1) # ca) \ (J(nl,1,1) # cl)

. mata: st_matrix("cmat",cmat)
```

# Implementation in Stata

## **Step 3:** Estimate the TVP-VAR model

```
. // Step 3: Estimate the TVP-VAR model
. tvpreg pi urate irate if time <= yq(2005,4), var varlag(1/2) level(90) ///
> cmatrix(cmat) chol nhor(0/20)
Running the Time-Varying-Parameter Estimation...
The model is:

       y(t) = [B(1,t),...,B(p,t),c(t)] × [y(t-1)',...,y(t-p)',1]' + e(t)
  Bt(L)y(t) = c(t) + u(t) = c(t) + Θ(0,t)ε(t)

 with lags (p) includes 1 2,
      dependent variable  y(t) (3×1): pi urate irate,
      B(t) = [B(1,t),...,B(p,t),c(t)],
      e(t) ~ N(0,Σ(e,t)), and A(t) × Σ(e,t) × A(t)' = Σ(ε,t) × Σ(ε,t)',


             _                          _
            |     1          0         0  |
   A(t) =   |  a(21,t)       1         0  |
            |_ a(31,t)    a(32,t)      1 _|

and Σ(ε,t) = diag[σ(1,t), σ(2,t), σ(3,t)].

The parameter is [vec(B(t)')',a(t)',lnσ(t)']',

 with a(t) = [a(21,t),a(31,t),a(32,t)]', and
      σ(t) = [σ(1,t), σ(2,t), σ(3,t)]'.

The constant parameter model is estimated by VAR.
```

# Implementation in Stata

## **Step 4:** Plot IRFs for different periods (1975Q1, 1981Q1, 1996Q1)

```
. // Step 4: Plot the estimation results
. gen period = 0

. replace period = 1 if time == yq(1975,1) | time == yq(1981,1) | ///
> time == yq(1996,1) // Prepare time indicator for figures
(3 real changes made)

. tvpplot, plotvarirf(pi:pi pi:urate pi:irate urate:pi urate:urate ///
> urate:irate irate:pi irate:urate irate:irate) plotconst period(period) ///
> name(figure1_1) periodlegend(1975Q1, 1981Q1, 1996Q1) tvpcolor(gray)
Plotting the impulse response function...
slope para: effect of pi on pi
slope para: effect of urate on pi
slope para: effect of irate on pi
slope para: effect of pi on urate
slope para: effect of urate on urate
slope para: effect of irate on urate
slope para: effect of pi on irate
slope para: effect of urate on irate
slope para: effect of irate on irate

. tvpplot, plotcoef(pi:L.pi pi:L.urate pi:L.irate urate:L.pi urate:L.urate ///
> urate:L.irate irate:L.pi irate:L.urate irate:L.irate) name(figure1_2) ///
> tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of L.pi on pi
slope para: effect of L.urate on pi
slope para: effect of L.irate on pi
slope para: effect of L.pi on urate
slope para: effect of L.urate on urate
slope para: effect of L.irate on urate
slope para: effect of L.pi on irate
slope para: effect of L.urate on irate
slope para: effect of L.irate on irate

. tvpplot, plotcoef(l1) name(figure1_3) tvpcolor(gray) ///
> title("Log standard deviation of shocks in inflation equation")
```

# Implementation in Stata



# Implementation in Stata

**Step 5:** Save the estimation results as new variables

```
. // Step 5: Save the estimation results as new variables
. predict pihat uhat ihat, xb y(pi urate irate)

. predict pires ures ires, residual y(pi urate irate)

. predict coef_pi_l1urate, coef(pi:L.urate)

. predict irf1_pi_urate, varirf(pi:urate) h(1)
```

# Example 2: TVP-LP

# Empirical setup

**Research question:** Examines fiscal shocks and their impact on government spending, see Ramey and Zubiary (2018), Inoue, Rossi, and Wang (2024b).

**Shock of Interest:** A one-unit military news shock ($\epsilon_{f,t}$) and its effect on government spending ($g_t$).

**TVP-LP:**

$$g_{t+h} = \alpha_{t+h} + \beta_{h,t+h}\epsilon_{f,t} + \sum_{j=1}^{4} \psi'_{j,t+h}\mathbf{w}_{t-j} + \xi_{t+h},$$

where

- $\mathbf{w}_t = (g_t, y_t, \epsilon_{f,t})'$
- $\beta_{h,t+h}$: Time-varying impulse response

# Implementation in Stata

**Step 1:** Import the data
**Step 2:** Create the C grid

```
.
. * Estimator II: TVP-LP (Government spending to a one-unit news shock)
. // Step 1:  Import the data
. use data_Fiscal.dta, clear

. tsset time

Time variable: time, 1889q4 to 2015q4
        Delta: 1 quarter

. // Step 2: Create the C grid
. mat define cmat = (0,3,6,9,12,15)
```

# Implementation in Stata

## **Step 3:** Estimate the TVP-LP model

```
. // Step 3: Estimate the TVP-LP model
. tvpreg gs shock gs_l* gdp_l* shock_l*, newey cmatrix(cmat) nhor(0/19) chol ///
> getband
Running the Time-Varying-Parameter Estimation...
The model is:

    y(t+h) = B(h,t+h) × x(t) + e(t+h)

 with horizon (h) includes 0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19,
      dependent variable   y(t) (1×1): gs,
      independent variable x(t) (14×1): shock gs_l1 gs_l2 gs_l3 gs_l4 gdp_l1 gdp_l2 gdp_l3 gdp_l4
shock_l1 shock_l2 shock_l3 shock_l4 _cons,
      e(t+h) ~ N(0,σ(e,t+h)^2).

The parameter is [vec(B(h,t+h)')',lnσ(e,t+h)]',

The constant parameter model is estimated by newey
```

# Implementation in Stata

## **Step 4 and 5:** Postestimation

```
. // Step 4: Plot the estimation results
. tvpplot, plotcoef(gs:shock) plotconst name(figure2_1) tvpcolor(gray)
Plotting the parameter path over horizons...
slope para: effect of shock on gs

. tvpplot, plotcoef(gs:shock) period(recession) name(figure2_2) tvpcolor(gray)
Plotting the parameter path over horizons...
slope para: effect of shock on gs

. tvpplot, plotcoef(gs:shock) plotnhor(1) plotconst name(figure2_3) ///
> tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of shock on gs

. tvpplot, plotcoef(gs:shock) plotnhor(1) period(recession) name(figure2_4) ///
> tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of shock on gs

. // Step 5: Save the estimation results as new variables
. predict gshat, xb h(0)

. predict gsres1, residual h(1)

. predict coef2_gs_shock, coef(gs:shock) h(2)
```

# Implementation in Stata




(a) Responses in all periods




(b) Responses in recessions




(c) Response across time




(d) Highlighting recessions

# Example 3: TVP-LP-IV

# Empirical setup

**Research question:** Estimate time-varying fiscal multiplier, see Ramey and Zubiary (2018), Inoue, Rossi, and Wang (2024b).

**TVP-LP-IV:**
The (cumulative) multipliers ($m_{h,t+h}$) are obtained as follows:

$$\sum_{j=0}^{h} y_{t+j} = \mu_{y,t+h} + m_{h,t+h} \sum_{j=0}^{h} g_{t+j} + \phi'_{y,t+h}(L)\mathbf{w}_t + v_{y,t+h}, \quad h = 0, 1, \ldots,$$

where

- $\sum_{j=0}^{h} g_{t+j}$ is endogenous and the news shock $\epsilon_{f,t}$ is used as an external instrument
- $m_{h,t+h}$: Time-varying multipliers

# Implementation in Stata

**Step 1:** Import the data

**Step 2:** Create the C grid

```
.
. * Estimator III: TVP-LP-IV (Fiscal multiplier)
. // Step 1:  Import the data
. use data_Fiscal.dta, clear

. tsset time

Time variable: time, 1889q4 to 2015q4
        Delta: 1 quarter

. // Step 2: Create the C grid
. * Input correct # of parameters
. mata: ny = 1; nx = 1; nz1 = 13; nz2 = 1

. * Input C grids for different blocks of parameters
. mata: cB = (3*(0::5))'; cv = (3*(0::5))'

. * Automatic calculations
. mata: nz = nz1 + nz2

. mata: nB = nx * nz + ny * nx + ny * nz1

. mata: nv = (nx + ny) * (nx + ny + 1) / 2

. mata: ncB = cols(cB); ncv = cols(cv)

. mata: cB = cB # J(1,ncv,1); cv = J(1,ncB,1) # cv

. mata: cmat = (J(nB,1,1) # cB) \ (J(nv,1,1) # cv)

. mata: st_matrix("cmat",cmat)
```

# Implementation in Stata

## **Step 3:** Estimate the TVP-LP-IV model

```
. // Step 3: Estimate the TVP-LP-IV model
. tvpreg gdp gs_l* gdp_l* shock_l* (gs = shock), cmatrix(cmat) nwlag(8) ///
> nhor(4/20) cum
Running the Time-Varying-Parameter Estimation...
The structural model is:

    cumy(t+h) = B(x,h,t+h) × cumx(t+h) + B(z1,h,t+h) × z(1,t) + v(2,t+h)

 with dependent variable   y(t) (1×1): gdp, cumy(t+h) = y(t)+...+y(t+h),
      endogeneous variable x(t) (1×1): gs, cumx(t+h) = x(t)+...+x(t+h), and

    cumx(t+h) = Π(2,h,t+h) × z(2,t) + Π(1,h,t+h) × z(1,t) + v(1,t+h)

 with included instruments z(1,t) (13×1): gs_l1 gs_l2 gs_l3 gs_l4 gdp_l1 gdp_l2 gdp_l3 gdp_l4
shock_l1 shock_l2 shock_l3 shock_l4 _cons,
      excluded instruments z(2,t) (1×1): shock.

The multivariate system is:

    _            _     _                                        _     _      _     _          _
   | cumx(t+h)    |   |                  Π(h,t+h)                |   | z(2,t)  |   | v(1,t+h)   |
   |              | = |                                          | × |         | + |            |
   |_ cumy(t+h)  _|   |_ B(x,h,t+h)×Π(h,t+h) + [0 B(z1,h,t+h)]  _|   |_ z(1,t) _|   | v(2,t+h)  _|

 with Π(h,t+h) = [Π(2,h,t+h),Π(1,h,t+h)], v(t+h) = [v(1,t+h)',v(2,t+h)']',
      v(t+h) ~ N(0,Σ(v,t+h)), Σ(v,t+h) is a symmetric matrix.

The parameter is [vec(Π(h,t+h)')',vec(B(x,h,t+h)')',vec(B(z1,h,t+h)')',vech(Σ(v,t+h))']'.

The constant parameter model is estimated by 2SLS.
```

# Implementation in Stata

## **Step 4 and 5:** Postestimation

```
. // Step 4: Plot the estimation results
. tvpplot, plotcoef(gdp:gs) period(recession) name(figure3_1) tvpcolor(gray)
Plotting the parameter path over horizons...
slope para: effect of gs on gdp

. tvpplot, plotcoef(gdp:gs) plotnhor(8) name(figure3_2) tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of gs on gdp

. // Step 5: Save the estimation results as new variables
. predict gdphat4, xb h(4) y(gdp)

. predict gsres8, residual h(8) y(gs)

. predict coef4_gdp_gs, coef(gdp:gs) h(4)
```

# Implementation in Stata




(a) Multipliers across horizons in recessions

(b) Two-year cumulative multiplier across time

Figure: Cumulative spending multipliers across time and across horizons.

# Example 4: TVP-weak IV

# Empirical setup

**Research question:** Estimate Phillips curve in unstable environment.

Consider the time-varying parameter hybrid Phillips curve in Inoue, Rossi, and Wang (2024a):

$$\pi_t = c_t + \lambda_t x_t + \gamma_{f,t}\pi_{t+1} + \gamma_{b,t}\pi_{t-1} + u_t,$$

where

- $x_t$ and $\pi_{t+1}$ are endogenous.
- The parameters of interest are $\lambda_t$, $\gamma_{f,t}$, and $\gamma_{b,t}$.

# Implementation in Stata

```
.
. * Estimator IV: TVP-weakIV (Phillips curve with weak instruments)
. // Step 1:  Import the data
. use data_PC_weak.dta, clear

. tsset time

Time variable: time, 1971q1 to 2018q4
        Delta: 1 quarter

. // Step 2: Create the C grid
. mat define cmat = (0, 1, 2, 3, 4, 5)

. // Step 3: Estimate the TVP-weakLV model
. tvpreg pi pib (x pif = x_l*), weakiv cmatrix(cmat) level(90) nwlag(19) ///
>  getband nodisplay
Running the Time-Varying-Parameter Estimation...

. // Step 4: Plot the estimation results
. tvpplot, plotcoef(pi:x) movavg(7) name(figure4_1) tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of x on pi

. tvpplot, plotcoef(pi:pif) movavg(7) name(figure4_2) tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of pif on pi

. tvpplot, plotcoef(pi:pib) movavg(7) name(figure4_3) tvpcolor(gray)
Plotting the parameter path over time...
slope para: effect of pib on pi

. // Step 5: Save the estimation results as new variables
. predict pihat, xb

. predict coef_pi_x, coef(pi:x)
```

# Implementation in Stata




(a) $\lambda_t$




(b) $\gamma_{f,t}$




(c) $\gamma_{b,t}$

Figure: The time-varying Phillips curve during the pandemic.

# Conclusions

WAR minimizing path estimators

- Müller-Petalas (2010): path estimator
- Inoue, Rossi, and Wang (2024b): TVP-LP and TVP-LP-IV
- Inoue, Rossi, and Wang (2024a): TVP-weak IV

- Inoue, Rossi, Wang, and Zhou (2024) "Parameter Path Estimation in Unstable Environments: The tvpreg Command"

# Thank You!